# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=shellerator
pkgver=34.c325e3d
pkgrel=4
pkgdesc='Simple command-line tool aimed to help pentesters quickly generate one-liner reverse/bind shells in multiple languages.'
arch=('any')
groups=('blackarch' 'blackarch-automation')
url='https://github.com/ShutdownRepo/shellerator'
license=('GPL3')
depends=('python' 'python-psutil' 'python-console-menu' 'python-argparse'
         'python-colorama' 'python-simple-term-menu')
makedepends=('git' 'python-setuptools')
source=("git+https://github.com/ShutdownRepo/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

build() {
  cd $pkgname

  python setup.py build
}

package() {
  cd $pkgname

  python setup.py install --root="$pkgdir" --prefix=/usr -O1 --skip-build
}

